# Strictly Monotone Brouwer Trees for Well Founded Recursion Over Multiple Values

Anonymous Author(s)

#### **Abstract**

Ordinals can be used to prove the termination of dependently typed programs. Brouwer trees are a particular ordinal notation that make it very easy to assign sizes to higher order data structures. They extend unary natural numbers with a limit constructor, so a function's size can be the least upper bound of the sizes of values from its image. These can then be used to define well founded recursion: any recursive calls are allowed so long as they are on values whose sizes are strictly smaller than the current size.

Unfortunately, Brouwer trees are not algebraically well behaved. They can be characterized equationally as a join-semilattice, where the join takes the maximum of two trees. However, this join does not interact well with the successor constructor, so it does not interact properly with the strict ordering used in well founded recursion.

We present Strictly Monotone Brouwer trees (SMB-trees), a refinement of Brouwer trees that are algebraically well behaved. SMB-trees are built using functions with the same signatures as Brouwer tree constructors, and they satisfy all Brouwer tree inequalities. However, their join operator distributes over the successor, making them suited for well founded recursion or equational reasoning.

This paper teaches how, using dependent pairs and careful definitions, an ill behaved definition can be turned into a well behaved one. Our approach is axiomatically lightweight: it does not rely on Axiom K, univalence, quotient types, or Higher Inductive Types. We implement a recursively-defined maximum operator for Brouwer trees that matches on successors and handles them specifically. Then, we define SMB-trees as the subset of Brouwer trees for which the recursive maximum computes a least upper bound. Finally, we show that every Brouwer tree can be transformed into a corresponding SMB-tree by joining it with itself an infinite number of times. All definitions and theorems are implemented in Agda.

*Keywords:* dependent types, Brouwer trees, well founded recursion

#### 1 Introduction

### 1.1 Recursion and Dependent Types

Dependently typed languages, such as Agda [?], Coq [Bertot and Castéran 2004], Idris [?] and Lean [?], bridge the gap between theorem proving and programming.

Functions defined in dependently typed languages are typically required to be *total*: they must provably halt in all inputs. Since the halting problem is undecidable, recursively-defined functions must be written in such a way that the type checker can mechanically deduce termination. Some functions only make recursive calls to structurally-smaller arguments, so their termination is apparent to the compiler. However, some functions cannot be easily expressed using structural recursion. For such functions, the programmer must instead use *well founded recursion*, showing that there is some ordering, with no infinitely-descending chains, for which each recursive call is strictly smaller according to this ordering. For example, the typical quicksort algorithm is not structurally recursive, but can use well founded recursion on the length of the lists being sorted.

#### 1.2 Ordinals

While numeric orderings work for first-order data, they are ill suited to recursion over higher-order data structures, where some fields contain functions.

There are many formulations of ordinals in dependent type theory, each with their own advantages and disadvantages.

#### 1.3 Contributions

This work defines *strictly monotone Brouwer Trees*, henceforth SMB-trees, a new presentation of ordinals that hit a sort of sweet-spot for defining functions by well founded recursion. Specifically, SMB-trees:

- are strictly ordered by a well founded relation;
- have a maximum operator which computes a leastupper bound;
- are *strictly-monotone* with respect to the maximum: if a < b and c < d, then  $\max a c < \max b d$ ;
- can compute the limits of arbitrary sequences;
- are light in axiomatic requirements: they are defined without using axiom K, univalence, quotient types, or higher inductive types.

#### 1.4 Uses for SMB-trees

**1.4.1 Well Founded Recursion.** Having a maximum operator for ordinals is particularly useful when traversing over multiple higher order data structures in parallel, where neither argument takes priority over the other. In such a case, a lexicographic ordering cannot be used.

As an example, consider a unification algorithm over some encoding of types, and suppose that  $\alpha$ -renaming or some

L

 other restriction prevents structural recursion from being used. To solve a unification problem  $\Sigma(x:A)$ .  $B=\Sigma(x:C)$ . D we must recursively solve A=C and  $\forall x. B[x]=D[x]$ . However, the type of x in the latter equation depends on the solution to the first equation, which is bounded by the size of the maximum of the sizes of both A and C. So for each recursive call to be on a smaller size, the size of a=c and b=d must both be strictly smaller than (a,b)=(c,d). In a lexicographic ordering where the size of the left-hand size dominates, we know that a is strictly smaller than (a,b), but we have no guarantees that TODO. Conversely, if we order unification problems by the size of the maximum of their two sides.

This style of well founded induction was used to prove termination in a syntactic model of gradual dependent types [?]. There, Brouwer trees were used to establish termination of recursive procedures for combining the type information in two imprecise types. The decreasing metric was the maximum size of the codes for the types being combined. Brouwer trees' arbitrary limits were used to assign sizes to dependent function and product types, and the strict monotonicity of the maximum operator was essential for proving that recursive calls were on strictly smaller arguments.

**1.4.2 Syntactic Models and Sized Types.** An alternate way view of our contribution is as a tool for modelling sized types [?]. The implementation of sized types in Agda has been shown to be unsound [?], due to the interaction between propositional equality and the top size  $\infty$  satisfying  $\infty < \infty$ . [Chan 2022] defines a dependently typed language with sized types that does not have a top size, proving it consistent using a syntactic model based on Brouwer trees.

SMB-trees provide the capability to extend existing syntactic models to sized types with a maximum operator. This brings the capability of consistent sized types closder to feature parity with Agda, which has a maximum operator for its sizes [?], while still maintaining logical consistency.

**1.4.3 Algebraic Reasoning.** Another advantage of SMB-trees is that they allow Brouwer trees to be interpreted using algebraic tools. SMB-trees can be described as In algebraic terminology, SMB-trees satisfy the following algebraic laws, up to the equivalence relation defined by  $s \approx t := s \le t \le s$ 

- Join-semlattice: the binary max is associative, commutative, and idempotent
- Bounded: there is a least tree Z such that  $\max t Z \approx t$
- Inflationary endomorphism: there is a successor operator  $\uparrow$  such that max  $(\uparrow t)$   $t \approx \uparrow t$  and  $\uparrow (\max s \ t) = \max(\uparrow s) \ (\uparrow t)$

Bezem and Coquand [2022] describe a polynomial time algorithm for solving equations in such an algebra, and describe its usefulness for solving constraints involving universe levels in dependent type checking. While equations involving limits of infinite sequences are undecidable, the inflationary laws could be used to automatically discharge some equations involving sizes. This algebraic presentation is particularly amenable to solving equations using free extensions of algebras [Allais et al. 2023; Corbyn 2021].

#### 1.5 Implementation

We have implemented SMB-trees in Agda 2.6.4. Our library specifically avoids Agda-specific features such as cubcal type theory or Axiom K, so we expect that the library can be easily ported to other proof assistants.

This paper is written as a literate Agda document, and the definitions given in the paper are valid Agda code. Several definitions are presented with their body omitted due to space restrictions. The full implementation can be found in the supplementary materials section of this submission.

#### 2 Brouwer Trees: An Introduction

Brouwer trees are a simple but elegant tool for proving termination of higher-order procedures. Traditionally, they are defined as follows:

```
data SmallTree : Set where
Z : SmallTree
↑ : SmallTree → SmallTree
Lim : (N → SmallTree) → SmallTree
```

Under this definition, a Brouwer tree is either zero, the successor of another Brouwer tree, or the limit of a countable sequence of Brouwer trees. However, these are quite weak, in that they can only take the limit of countable sequences. To represent the limits of uncountable sequences, we can paramterize our definition over some Universe à la Tarski:

```
module Brouwer \{\ell\}

(C : \text{Set } \ell)

(El : C \to \text{Set } \ell)

(CN : C) (CNIso : \text{Iso } (El \ CN) \ \mathbb{N}) where
```

Our module is paramterized over a universe level, a type  $\mathbb C$  of codes, and an "elements-of" interpretation function El, which computes the type represented by each code. We require that there be a code whose interpretation is isomorphic to the natural numbers, as this is essential to our construction in  $\ref{eq:construction}$ . Increasingly larger trees can be obtained by setting  $\mathbb C := \operatorname{Set} \ell$  and El := id for increasing  $\ell$ . However, by defining an inductive-recursive universe, one can still capture limits over some non-countable types, since Tree is in Set whenever  $\mathbb C$  is.

We then generalize limits to any function whose domain is the interpretation of some code.

```
data Tree : Set \ell where
Z : Tree
↑ : Tree → Tree
Lim : \forall (c : C) \rightarrow (f : El \ c \rightarrow Tree) \rightarrow Tree
```

The small limit constructor can be recovered from the natural-number code

Brouwer trees are a the quintessential example of a higherorder inductive type. 1: Each tree is built using smaller trees or functions producing smaller trees, which is essentially a way of storing a possibly infinite number of smaller trees.

#### 2.1 Ordering Trees

Our ultimate goal is to have a well-founded ordering<sup>2</sup>, so we define a relation to order Brouwer trees.

```
data \_ \le \_ : Tree \longrightarrow Tree \longrightarrow Set \ell where

\le -Z : \forall \{t\} \longrightarrow Z \le t

\le -\text{sucMono} : \forall \{t_1 \ t_2\}

\longrightarrow t_1 \le t_2

\longrightarrow \uparrow t_1 \le \uparrow t_2

\le -\text{cocone} : \forall \{t\} \{c : C\} (f : El \ c \longrightarrow \text{Tree}) (k : El \ c)

\longrightarrow t \le f \ k

\longrightarrow t \le \text{Lim} \ c \ f

\le -\text{limiting} : \forall \{t\} \{c : C\}

\longrightarrow (f : El \ c \longrightarrow \text{Tree})

\longrightarrow (\forall k \longrightarrow f \ k \le t)

\longrightarrow \text{Lim} \ c \ f \le t
```

This relation is reflexive:

```
\leq-refl : \forall t \rightarrow t \leq t

\leq-refl Z = \leq-Z

\leq-refl (\uparrow t) = \leq-sucMono (\leq-refl t)

\leq-refl (Lim c f)

= \leq-limiting f (\lambda k \rightarrow \leq-cocone f k (\leq-refl (f k)))
```

Crucially, it is also transitive, making the relation a preorder. We modify our the order relation from that of Kraus et al. [2023] so that transitivity can be proven constructively, rather than adding it as a constructor for the relation. This allows us to prove well-foundedness of the relation without needing quotient types or other advanced features.

```
\leq-trans: \forall \{t_1 \ t_2 \ t3\} \rightarrow t_1 \leq t_2 \rightarrow t_2 \leq t3 \rightarrow t_1 \leq t3

\leq-trans \leq-Z p23 = \leq-Z

\leq-trans (\leq-sucMono p12) (\leq-sucMono p23)

= \leq-sucMono (\leq-trans p12 \ p23)

\leq-trans p12 (\leq-cocone f \ k \ p23)

= \leq-cocone f \ k \ (\leq-trans p12 \ p23)

\leq-trans (\leq-limiting f \ x) \ p23
```

```
= \le -\liminf f \ (\lambda \ k \to \le -\operatorname{trans} (x \ k) \ p23) 
\le -\operatorname{trans} (\le -\operatorname{cocone} f \ k \ p12) \ (\le -\liminf g \ f \ x) 
= \le -\operatorname{trans} p12 \ (x \ k) 
278
279
```

We create an infix version of transitivity for more readable construction of proofs:

```
\_ \le \ ^\circ \_ : \forall \ \{t_1 \ t_2 \ t3\} \longrightarrow t_1 \le t_2 \longrightarrow t_2 \le t3 \longrightarrow t_1 \le t3
 lt1 \le \ ^\circ _0 \ lt2 = \le \text{-trans} \ lt1 \ lt2
```

**2.1.1 Strict Ordering.** We can define a strictly-less-than relation in terms of our less-than relation and the successor constructor:

```
\_<\_: \mathsf{Tree} \to \mathsf{Tree} \to \mathsf{Set}\ \ell
t_1 < t_2 = \uparrow t_1 \le t_2
```

 $\leq \uparrow t : \forall t \longrightarrow t \leq \uparrow t$ 

That is, a  $t_1$  is strictly smaller than  $t_2$  if the tree one-size larger than  $t_1$  is as small as  $t_2$ . This relation has the properties one expects of a strictly-less-than relation: it is a transitive sub-relation of the less-than relation, every tree is strictly less than its successor, and no tree is strictly smaller than zero.  $\boxed{\text{JE}} \hspace{-0.2cm} \blacktriangleright \text{TODO more?} \blacktriangleleft$ 

# 2.2 Well Founded Induction

Recall the definition of a constructive well founded relation:

```
data Acc \{A : Set \ a\} (\_<\_ : A \rightarrow A \rightarrow Set \ \ell) \ (x : A) : Set \ (a \boxtimes \ell) \ where acc : (rs : \forall \ y \rightarrow y < x \rightarrow Acc \ \_<\_ y) \rightarrow Acc \ \_<\_ x

321

WellFounded : (A \rightarrow A \rightarrow Set \ \ell) \rightarrow Set \ \_

WellFounded \_<\_ = \forall \ x \rightarrow Acc \ \_<\_ x

323

324
```

That is, an element of a type is accessible for a relation if all strictly smaller elements of it are also accessible. A relation is well founded if all values are accessible with respect to that relation. This can then be used to define induction with arbitrary recursive calls on smaller values:

<sup>&</sup>lt;sup>1</sup>Not to be confused with Higher Inductive Types (HITs) from Homotopy Type Theory [Univalent Foundations Program 2013]

<sup>&</sup>lt;sup>2</sup>Technically, this is a well-founded quasi-ordering because there are pairs of trees which are related by both  $\leq$  and  $\geq$ , but which are not propositionally equal.

```
wfRec: (P : A \rightarrow \mathsf{Set} \ \ell)

\rightarrow (\forall \ x \rightarrow ((y : A) \rightarrow y < x \rightarrow P \ y) \rightarrow P \ x)

\rightarrow \forall \ x \rightarrow P \ x
```

Following the construction of Kraus et al. [2023], we can show that the strict ordering on Brouwer trees is well founded. First, we prove a helper lemma: if a value is accessible, then all (not necessarily strictly) smaller terms are are also accessible.

```
smaller-accessible : (x : Tree)

\rightarrow Acc \le x \rightarrow \forall y \rightarrow y \le x \rightarrow Acc \le y

smaller-accessible x (acc r) y x < y

= acc (\lambda y' y' < y \rightarrow r y' (< \le -in < y' < y x < y))
```

Then we use structural reduction to show that all terms are accesible. The key observations are that zero is trivially accessible, since no trees are strictly smaller than it, and that the only way to derive  $\uparrow t \leq (\text{Lim } c f)$  is with  $\leq$ -cocone, yielding a concrete index k for which  $\uparrow t \leq f k$ , on which we can recur.

```
ordWF : WellFounded \_<\_
ordWF Z = acc \ \lambda \_()
ordWF (\uparrow x)
= acc \ (\lambda \{ y (\le -sucMono \ y \le x) \rightarrow smaller-accessible \ x (ordWF \ x) \ y \ y \le x \})
ordWF (Lim c \ f) = acc \ helper
where
helper : (y : Tree) \rightarrow (y < Lim \ c \ f)
\rightarrow Acc \ _< \ y
helper \ y \ (\le -cocone \ .f \ k \ y < fk)
= smaller-accessible \ (f \ k)
(ordWF \ (f \ k)) \ y \ (< -in - \le y < fk)
```

# 3 First Attempts at a Join

In this section, we present two faulty implmentations of a join operator for trees. The first uses limits to define the join, but does not satisfy strict monotonicity. The second is defined inductively. Its satisfies strict monotonicity, but fails to be the least of all upper bounds, and requires us to assume that limits are only taken over non-empty types. In ??, we define SMB-trees a refinement of Brouwer trees that combines the benefits of both versions of the maximum.

#### 3.1 Limit-based Maximum

Since the limit constructor finds the least upper bound of the image of a function, it should be possible to define the maximum of two trees as a special case of general limits. Indeed, we can compute the maximum of  $t_1$  and  $t_2$  as the limit of the function that produces  $t_1$  when given 0 and  $t_2$  otherwise.

```
limMax : Tree \rightarrow Tree \rightarrow Tree
limMax t_1 t_2 = NLim \lambda n \rightarrow if0 n t_1 t_2
```

 $\lim Max \le L : \forall \{t_1 \ t_2\} \longrightarrow t_1 \le \lim Max \ t_1 \ t_2$ 

 $\lim Max \leq L \{t_1\} \{t_2\}$ 

This version of the maximum has several of the properties we want from a maximum function: it is monotone, idempotent, commutative, and is a true least-upper-bound of its inputs.

JE ► TODO update description From these properties, we can compute several other useful properties: monotonicity, commutativity, and that it is in fact the least of all upper bounds.

```
\begin{aligned} & \mathsf{limMaxMono} : \forall \left\{ t_1 \ t_2 \ t_1' \ t_2' \right\} \\ & \longrightarrow t_1 \le t_1' \longrightarrow t_2 \le t_2' \\ & \longrightarrow \mathsf{limMax} \ t_1 \ t_2 \le \mathsf{limMax} \ t_1' \ t_2' \end{aligned}
```

 $limMaxCommut : \forall \{t_1 \ t_2\} \rightarrow limMax \ t_1 \ t_2 \leq limMax \ t_2 \ t_1$ 

```
limMaxLUB : \forall \{t_1 \ t_2 \ t\} \rightarrow t_1 \le t \rightarrow t_2 \le t \rightarrow limMax \ t_1 \ t_2 \le t
```

It is not surprising that this version of the maximum is a least upper bound: by definition Lim computes the least upper bound of a function's image, and limMax is simply Lim applied to a function whose image has (at most) two elements.

**3.1.1 Limitation: Strict Monotonicity.** The one crucial property that this formulation lacks is that it is not strictly monotone: we cannot deduce  $\max t_1 \ t_1 < \max t_1' \ t_2'$  from  $t_1 < t_1'$  and  $t_2 < t_2'$ . This is because the only way to construct a proof that  $\uparrow t \leq \lim c f$  is using the  $\leq$ -cocone constructor. So we would need to prove that  $\uparrow (\max t_1 \ t_2) \leq t_1'$  or that  $\uparrow (\max t_1 \ t_2) \leq t_2'$ , which cannot be deduced from the premises alone. What we want is to have  $\uparrow \max (t_1) \ t_2 \leq \max (\uparrow t_1) \ (\uparrow t_2)$ , so that strict monotonicity is a direct consequence of ordinary monotonicity of the maximum. This is not possible when defining the constructor as a limit.

497

498

499

500

501

502

503

504

505

506

507

508

509

510

511

512

513

514

515

516

517

518

519

520

521

522

523

524

525

526

527

529

530

531

532

533

534

535

536

537

538

539

540

541

542

543

544

545

546

547

548

549

550

#### 3.2 Recursive Maximum

441

442

443

444

445

446

447

448

449

450

451

452

453

454

455

476

477

478

479

480

481

482

483

484

485

486

487

488

489

490

491

492

493

494

495

In our next attempt at defining a maximum operator, we obtain strict monotonicity by making indMax  $(\uparrow t_1)$   $(\uparrow t_2)$  =  $\uparrow$  (indMax  $t_1$   $t_2$ ) hold definitionally. Then, provided indMax is monotone, it will also be strictly monotone.

To do this, we compute the maximum of two trees recursively, pattern matching on the operands. We use a *view* [?] datatype to identify the cases we are matching on: we are matching on two arguments, which each have three possible constructors, but several cases overlap. Using a view type lets us avoid enumerating all nine possibilities when defining the maximum and proving its properties.

To begin, we parameterize our definition over a function yielding some element for any code's type.

```
module IndMax {ℓ}
456
            (C: \mathbf{Set} \ \ell)
457
            (El: \mathbb{C} \longrightarrow \operatorname{Set} \ell)
458
            (CN : C) (CNIso : Iso (El CN) N)
459
             (default : (c : \mathbb{C}) \rightarrow El \ c) where
461
             We then define our view type:
462
463
         private
464
             data IndMaxView : Tree \rightarrow Tree \rightarrow Set \ell where
465
                IndMaxZ-L : \forall \{t\} \rightarrow IndMaxView Z t
466
                IndMaxZ-R : \forall \{t\} \rightarrow IndMaxView \ t \ Z
467
                IndMaxLim-L : \forall \{t\} \{c : C\} \{f : El \ c \rightarrow Tree\}
468
                   \rightarrow IndMaxView (Lim c f) t
469
                IndMaxLim-R : \forall \{t\} \{c : C\} \{f : El \ c \rightarrow \mathsf{Tree}\}\
470
                   \rightarrow (\forall \{c' : C\} \{f' : El \ c' \rightarrow \mathsf{Tree}\} \rightarrow \neg (t \equiv \mathsf{Lim} \ c' f'))
471
                   \rightarrow IndMaxView t (Lim c f)
472
                IndMaxLim-Suc : \forall \{t_1 \ t_2\} \rightarrow \text{IndMaxView} (\uparrow t_1) (\uparrow t_2)
473
         opaque
474
475
            indMaxView : \forall t_1 t_2 \rightarrow IndMaxView t_1 t_2
```

Our view type has five cases. The first two handle when either input is zero, and the second two handle when either input is a limit. The final case is when both inputs are successors. *indMaxView* computes the view for any pair of trees.

The maximum is then defined by pattern matching on the view for its arguments:

```
indMax : Tree \rightarrow Tree \rightarrow Tree
indMax': \forall \{t_1 \ t_2\} \rightarrow IndMaxView \ t_1 \ t_2 \rightarrow Tree
indMax t_1 t_2 = indMax' (indMaxView t_1 t_2)
indMax' \{.Z\} \{t_2\} IndMaxZ-L = t_2
indMax' \{t_1\} \{.Z\} IndMaxZ-R = t_1
indMax' \{(Lim \ c \ f)\} \{t_2\} IndMaxLim-L
  = Lim c \lambda x \rightarrow \text{indMax}(f x) t_2
indMax' \{t_1\} \{(Lim \ c \ f)\} (IndMaxLim-R )
  = Lim c (\lambda x \rightarrow \text{indMax } t_1 (f x))
indMax' \{(\uparrow t_1)\} \{(\uparrow t_2)\} IndMaxLim-Suc = \uparrow (indMax t_1 t_2)
```

The maximum of zero and t is always t, and the maximum of *t* and the limit of *f* is the limit of the function computing the maximum between t and f x. Finally, the maximum of two successors is the successor of the two maxima, giving the definitional equality we need for strict monotonicity.

This definition only works when limits of all codes are inhabited. The  $\leq$ -limiting constructor means that Lim c  $f \leq$ Z whenever *El c* is uninhabited. So indMax  $\uparrow$ Z Lim *c f* will not actually be an upper bound for  $\uparrow Z$  if c has no inhabitants. In ?? we show how to circumvent this restriction.

Under the assumption that all code are inhabited, we obtain several of our desired properties for a maximum: it is an upper bound, it is monotone and strictly monotonicity, and it is associative and commutative.

```
opaque
   unfolding indMax indMax'
  indMax-\leq L: \forall \{t_1 \ t_2\} \rightarrow t_1 \leq indMax \ t_1 \ t_2
  indMax-\leq L\{t_1\}\{t_2\} with indMaxView\ t_1\ t_2
  ... | IndMaxZ-L = \leq -Z
  ... | IndMaxZ-R = \leq-refl
  ... | IndMaxLim-L \{f = f\}
      = extLim f(\lambda x \rightarrow \text{indMax}(f x) t_2)(\lambda k \rightarrow \text{indMax-} \leq L)
  ... | IndMaxLim-R \{f = f\}
      = underLim \lambda k \rightarrow \text{indMax-} \{t_2 = f k\}
   ... | IndMaxLim-Suc
      = ≤-sucMono indMax-≤L
  indMax - \le R : \forall \{t_1 \ t_2\} \longrightarrow t_2 \le indMax \ t_1 \ t_2
   -- Symmetric
  indMax-monoL : \forall \{t_1 \ t'_1 \ t_2\}
      \rightarrow t_1 \le t_1' \rightarrow \text{indMax } t_1 \ t_2 \le \text{indMax } t_1' \ t_2
  indMax-monoR : \forall \{t_1 \ t_2 \ t_2'\}
      \rightarrow t_2 \le t_2' \rightarrow \text{indMax } t_1 \ t_2 \le \text{indMax } t_1 \ t_2'
  indMax-mono : \forall \{t_1 \ t_2 \ t'_1 \ t'_2\}
      \rightarrow t_1 \le t_1' \rightarrow t_2 \le t_2' \rightarrow \text{indMax } t_1 \ t_2 \le \text{indMax } t_1' \ t_2'
  indMax\text{-strictMono}: \forall \{t_1 \ t_2 \ t_1' \ t_2'\}
      \rightarrow t_1 < t_1' \rightarrow t_2 < t_2' \rightarrow \text{indMax } t_1 \ t_2 < \text{indMax } t_1' \ t_2'
  indMax-strictMono lt1 lt2 = indMax-mono lt1 lt2
   indMax-assocL : \forall t_1 t_2 t_3
      \rightarrow indMax t_1 (indMax t_2 t3) \leq indMax (indMax t_1 t_2) t3
   indMax-assocR : \forall t_1 t_2 t_3
      \rightarrow indMax (indMax t_1 t_2) t3 \le indMax t_1 (indMax t_2 t3)
  indMax-commut : \forall t_1 t_2
      \rightarrow indMax t_1 t_2 \le indMax t_2 t_1
```

552

553

554

555

556

557

558

559

560

561

562

563

564

565

566

567

568

569

570

571

572

573

574

575

576

577

578

579

580

581

582

583

584

585

586

587

588

589

590

591

592

593

594

595

596

597

598

599

600

601

602

603

604

605

607

609

610

611

612

613

614

622

623

624

625

626

627

628

629

630

631

632

633

634

635

636

637

638

639

640

641

642

643

645

646

647

648

649

650

651

652

653

654

655

656

657

658

659

660

**3.2.1** Limitation: Idempotence. The problem with an inductive definition of the maximum is that we cannot prove that it is idempotent. Since indMax is associative and commutative, proving idempotence is equivalent to proving that it computes a true least-upper-bound.

The difficulty lies in showing that  $\operatorname{indMax}(\operatorname{Lim} c f)(\operatorname{Lim} c f) \leq \operatorname{indMax} t = \operatorname{NLim}(\lambda n \to \operatorname{nindMax} t n)$ (Lim c f). By our definition, indMax (Lim c f) (Lim c f) reduces to

```
(\operatorname{Lim} c \lambda x \to (\operatorname{Lim} c \lambda y \to \operatorname{indMax} (f x) (f y))) \le \operatorname{Lim} c f
```

We cannot use ≤-cocone to prove this, since the left hand side is not necessarily equal to f k for any k : El c. So the only possibility is to use ≤-limiting. Applying it twice, along with a use of commutatativity of indMax, we are left with the following goal:

```
(\forall x \to (\forall y \to \text{indMax} (f x) (f y))) \le \text{Lim } c f
```

There is no a priori way to prove this goal without already having a proof that indMax is a least upper bound. But proving that was the whole point of proving idempotence! An inductive hypothesis would give that  $indMax(f x)(f x) \le f(x)$  $f x \leq Lim \ c \ f$ , but it does not apply when the arguments to indMax are not equal. Because we are working with constructive ordinals, we have no trichotomy property [?], and hence no guarantee that indMax(f x)(f y) will be one of f x and f y.

We now have two competing defintiions for the maximum: the limit version, which is not strictly monotone, and the inductive version, which is not actually a least upper bound. In the next section, we describe a large class of trees for which indMax is idempotent, and hence does compute a true upper bound. We then use that in ?? to create a version of ordinals whose join has the best properties of both limMax and indMax. JE ▶TODO recall the algebraic definition of semilattice◀

# Trees with a Strictly-Monotone **Idempotent Join**

#### 4.1 Well-Behaved Trees

Our first step in defining an ordinal notation with a well behaved maximum is to identify a class of Brouwer trees which are well behaved with respect to the inductive maximum. As we saw in

The answer, it turns out, is more limits: if we indMax a term with itself an infinite number of times, the result will be idempotent with respect to indMax. First, we define a function to indMax a term with itself n times or a given number *n*:

```
nindMax : Tree \rightarrow \mathbb{N} \rightarrow Tree
nindMax t N.zero = Z
nindMax \ t \ (N.suc \ n) = indMax \ (nindMax \ t \ n) \ t
```

To compute a tree equivalent to the infinite chain of applications indMax t (indMax t (indMax t ...)), we take the limit of *n* applications over all *n*:

```
indMax \infty : Tree \rightarrow Tree
```

This operator has useful basic properties: it is monotone, and it computes an upper bound on is argument.

```
indMax \infty-self : \forall t \rightarrow t \le indMax \infty t
                                                                                                             615
                                                                                                             616
indMax∞-mono : \forall \{t_1 \ t_2\}
                                                                                                             617
   \rightarrow t_1 \leq t_2
                                                                                                             618
   \rightarrow (indMax\otimes t_1) \leq (indMax\otimes t_2)
                                                                                                             619
```

However, the most important property we want from indMax∞<sub>20</sub> is that indMax is idempotent with respect to it. The first step to showing this is realizing that we can take the maximum of t and  $indMax \infty$  t and we have a tree that is no larger than indMax∞ t: because it is already an infinite chain of applications, adding one more makes no difference.

```
indMax - \infty lt 1 : \forall t \rightarrow indMax (indMax \infty t) t \leq indMax \infty t
indMax-∞lt1 t = \le-limiting \lambda k \rightarrow helper (Iso.fun CNIso k)
   where
   helper: \forall n \rightarrow \text{indMax} (\text{nindMax } t n) t \leq \text{indMax} \infty t
   helper n =
      \leq-cocone _ (Iso.inv CNIso (N.suc n))
      (subst (\lambda sn \rightarrow \text{nindMax } t \text{ (}\mathbb{N}.\text{suc } n\text{)} \leq \text{nindMax } t \text{ sn)}
         (sym (lso.rightlnv CNIso (suc n)))
        (≤-refl ))
```

If adding one more indMax t has no effect, then adding nmore will also have no effect:

```
indMax-∞ltn : ∀ n t
  \rightarrow indMax (indMax \infty t) (nindMax t n) \leq indMax \infty t
indMax-\infty Itn N.zero t = indMax-\le Z (indMax \infty t)
indMax-\infty ltn (N.suc n) t =
  indMax-monoR (indMax-commut (nindMax t n) t)
  \leq \frac{\circ}{9} indMax-assocL (indMax\infty t) t (nindMax t n)
  \leq \frac{\circ}{\circ} \text{ indMax-monoL (indMax-<math>\inftylt1 t)}
  ≤ % indMax-∞ltn n t
```

By our inductive definition of indMax, we have that

```
indMax (indMax \infty t)(indMax \infty t)
```

is equal to

```
NLim (\lambda n \rightarrow \text{indMax} (\text{nIndMax} n t) (\text{indMax} \infty t))
```

Our previous lemma gives that, for any n, indMax $\infty$  t is an upper bound for indMax (nIndMax n t) (indMax $\infty t$ )). So  $\leq$ -limiting gives that the limit over all n is also bounded by  $\operatorname{indMax} \propto t$ , i.e. Lim constructs the least of all upper bounds. This gives us our key result: up to  $\leq$ , indMax is idempotent on values constructed with indMax∞.

717

718

719

721

723

724

725

726

727

728

729

730

731

732

735

736

737

738

740

741

742

743

744

745

746

747

748

755

756

757

758

759

760

761

762

763

764

765

766

767

768

769

770

```
indMax∞-idem : ∀ t
  \rightarrow indMax (indMax\infty t) (indMax\infty t) \leq indMax\infty t
indMax∞-idem t =
  \leq-limiting \lambda k \rightarrow
    (indMax-commut (nindMax t (Iso.fun CNIso k)) (indMax<math> \infty t))
  \leq \frac{9}{9} indMax-\inftyltn (Iso.fun CNIso k) t
```

661

662

663

664

665

666

667

668

669

670

671

672

673

674

675

676

677

678

679

680

681

682

683

684

685

686

687

688

689

690

691

692

693

694

696

697

698

699

700

701

702

703

704

705

706

707

708

709

710

711

712

713

714

715

There is one last property to prove that will be useful in the next section:  $indMax \infty t$  is a lower bound on t, and hence equivalent to it, whenever indMax is idempotent on t. If taking indMax of t with itself does not increase it size, doing so *n* times will not increase it size, so again the result follows from Lim being the least upper bound.

```
\operatorname{indMax} \infty - \leq : \forall \{t\} \longrightarrow \operatorname{indMax} t \ t \leq t \longrightarrow \operatorname{indMax} \infty t \leq t
\operatorname{indMax} \infty \le lt = \le \operatorname{-limiting} \lambda k \to \operatorname{nindMax} \le (\operatorname{Iso.fun} \operatorname{CNIso} k) lt_{Z} = \operatorname{MkTree} \operatorname{Brouwer.Z} \operatorname{Brouwer.S} - Z
   where
      nindMax-\le : \forall \{t\} \ n \rightarrow indMax \ t \ t \le t \rightarrow nindMax \ t \ n \le t
      nindMax \le N.zero lt = \le -Z
      nindMax = \{t = t\} (N.suc n) lt = (indMax - monoL)\{t_1 = nindMax + towert\}  constituting the limit of a sequence of SMB-
```

An immediate corollary of this is that  $indMax \infty (indMax \infty t)$ is equivalent to  $indMax \infty t$ .

# 4.2 Strictly Monotone Brouwer Trees

Now that we have identified a substantial class of well behaved Brouwer trees, we want to define a new type containing only those trees. These are SMB-trees: strictly monotone Brouwer trees. In this section, we will define them, and show how they can be given a similar interface to Brouwer trees.

To begin, we declare a new Agda module, with the same parameters we have been working with thus far: a type of codes, interpretations of those codes into types, and a code whose interpretation is isomorphic to  $\mathbb{N}$ .

```
module SMBTree {ℓ}
  (\mathcal{C}: \mathbf{Set} \ \ell)
  (El: \mathbb{C} \to \operatorname{Set} \ell)
  (CN : C) (CNIso : Iso (El CN) N) where
```

We import all of our definitions so far, using the "Brouwer" prefix to distinguish them from the trees and ordering we are about to define. Critically, we do not instantiate these with the same interpretation function. Instead, we interpret each code wrapped in Maybe. This ensures that we always take Brouwer limits over non-empty sets, an assumption that was critical for the definitions of ??. However, we place no such restriction on SMB-trees.

```
import Brouwer C (\lambda c \rightarrow \text{Maybe}(El \ c)) CN (maybeNatIso CNIso) as Brouwer Brouwer C (\lambda c \rightarrow \text{Maybe}(El \ c)) CN (maybeNatIso CNIso) as Brouwer
```

**4.2.1 Refining Brouwer Trees.** We define SMB-trees as a dependent record, containing an underlying Brouwer tree, and a proof that indMax is idempotent on this tree.

```
record SMBTree : Set ℓ where
 constructor MkTree
```

```
field
   rawTree: Brouwer.Tree
   isIdem: (indMax rawTree rawTree) Brouwer.≤ rawTree
open SMBTree
```

We can then define so-called "smart-constructors" corresponding to each of the constructors for Brouwer-trees: zero, successor, and limit. Zero and successor directly correspond to the Brouwer tree zero and successor. Their proofs of idempotence are trivial from the properties of Brouwer ≤.

```
unfolding indMax
```

```
\mathsf{Z}:\mathsf{SMBTree}
\uparrow: SMBTree \rightarrow SMBTree
```

↑ (MkTree t pf) = MkTree (Brouwer. ↑ t) (Brouwer. ≤ -sucMono pf)

trees is not so easy. Since we instantiated El to wrap its result in Maybe, we need to handle *nothing* for each limit, but we can use Z as a default value, since adding it to any sequence does not change the least upper bound. More challenging is how, as we saw in ??, Brouwer trees do not have  $\operatorname{indMax} (\operatorname{Lim} c f) (\operatorname{Lim} c f) \leq \operatorname{Lim} c f$ , so we cannot directly produce a proof of idempotence.

Our key insight is to define limits of SMB-trees using indMax∞ on the underlying trees: for any function producing SMB-trees, we take the limit of the underlying trees, then indMax that result with itself an infinite numer of times. The idempotence proof is then the property of indMax∞ that we proved in ??.

```
\mathsf{Lim} : \forall \ (c : \mathcal{C}) \to (f : El \ c \to \mathsf{SMBTree}) \to \mathsf{SMBTree}
                                                                                                              749
\operatorname{Lim} c f =
                                                                                                              750
                                                                                                              751
   MkTree
   (\text{indMax} \otimes (\text{Brouwer.Lim } c \text{ (maybe' } (\lambda x \rightarrow \text{rawTree } (f x)) \text{ Brouwer.Z}))
   (indMax∞-idem _)
                                                                                                              754
```

**4.2.2 Ordering SMB-trees.** SMB-trees are ordered by the order on their underlying Brouwer trees:

```
record \leq (t_1 t_2 : SMBTree) : Set \ell where
 constructor mk≤
 inductive
  field
```

Having a successor function allows us to define a strict ording on SMB-trees.

```
< : SMBTree \rightarrow SMBTree \rightarrow Set \ell
_{-}<_{-} t_1 t_2 = (\uparrow t_1) \le t_2
```

772

773

774

775

776

777

778

779

780

781

782

783

784

785

786

787

788

789

790

791

792

793

794

795

796

797

798

799

800

801

802

803

804

805

806

807

808

809

810

811

812

813

814

815

816

817

818

819

820

821

822

823

824

825

826

827

829

831

832

833

834

835

836

837

838

839

840

841

842

843

844

845

846

847

848

849

850

851

852

863

864

865

866

867

868

869

870

871

872

873

874

875

876

877

878

879

880

The next step is to prove that our SMB-tree constructors satisfy the same inequalities as Brouwer trees. Since SMBtrees are ordered by their underlying Brouwer trees, most properties can be directly lifted from Brouwer trees to SMB-

```
opaque
   unfolding Z 1
   \leq \uparrow : \forall t \longrightarrow t \leq \uparrow t
   \leq \uparrow t = mk \leq (Brouwer. \leq \uparrow t)
   \leq  ^{\circ}_{9}: \forall \{t_1 \ t_2 \ t3\} \rightarrow t_1 \leq t_2 \rightarrow t_2 \leq t3 \rightarrow t_1 \leq t3
   \leq \frac{9}{9} (mk\leq lt1) (mk\leq lt2) = mk\leq (Brouwer.\leq-trans lt1 lt2)
   \leq-refl: \forall \{t\} \longrightarrow t \leq t
   \leq-refl = mk\leq (Brouwer.\leq-refl )
```

The constructors for  $\leq$  each have a counterpart for SMBtrees. For zero and successor, these are trivially lifted.

```
\leq -Z : \forall \{t\} \longrightarrow Z \leq t
\leq -Z = mk \leq Brouwer. \leq -Z
\leq-sucMono : \forall \{t_1 \ t_2\} \rightarrow t_1 \leq t_2 \rightarrow \uparrow t_1 \leq \uparrow t_2
\leq-sucMono (mk\leq lt) = mk\leq (Brouwer.\leq-sucMono lt)
```

The constructors for ordering limits require more attention. To show that an SMB-tree limit is an upper-bound, we use the fact that the underlying limit was an upper bound, and the fact that indMax∞ is an upper bound, since the SMBtree Lim wraps its result in indMax∞. Note that, since we already have transitivity for our new  $\leq$ , we can simply show that f k is less than the limit of f, avoiding the more complicated form of  $\leq$ -cocone.

```
\leq-limUpperBound : \forall \{c : C\} \rightarrow \{f : El \ c \rightarrow \mathsf{SMBTree}\}
  \rightarrow \forall k \rightarrow f k \leq \text{Lim } c f
\leq-limUpperBound \{c = c\} \{f = f\} k
  = mk \le (Brouwer. \le -cocone _ (just k) (Brouwer. \le -refl _)
                Brouwer. \leq \frac{9}{9} ind Max \infty-self (Brouwer. Lim c ))
```

Finally, we need to show that the SMT-tree limit is less than all other upper bounds. Suppose t : SMBTRee is an upper bound for f, and  $t_u$  is the underlying tree for t, and  $f_u$ computes the underlying trees for f. Then  $\leq$ -limiting gives that the underlying tree for *t* is an upper bound for the trees underlying the image of f. However, the SMB-tree limit wraps its result in indMax∞. The monotonicity of indMax∞ then gives that  $\operatorname{indMax}(\operatorname{Lim} c f_u)$  is less than  $\operatorname{indMax} \infty t'$ . In ??, we showed that indMax on had no effect on Brouwer trees that indMax was idempotent on. This is exactly what the isIdem field of SMB-trees contains! So we have indMax $\infty$   $t' \leq$ t', and transitivity gives our result.

```
\leq-limLeast : \forall \{c : C\} \rightarrow \{f : El \ c \rightarrow \mathsf{SMBTree}\}
   \rightarrow {t : SMBTree}
   \rightarrow (\forall k \rightarrow f \ k \leq t) \rightarrow \text{Lim } c \ f \leq t
\leq-limLeast \{f = f\} \{t = MkTree \ t \ idem\} \ lt
```

```
= mk≤ (
  indMax∞-mono
     (Brouwer.≤-limiting _
        (maybe (\lambda k \rightarrow \text{get} \leq (lt \ k)) Brouwer.\leq -Z))
  Brouwer. \leq \frac{\circ}{9} (indMax \infty - \leq idem))
```

**4.2.3** The Join for SMB-trees. Our whole reason for defining SMB-trees was to define a well-behaved maximum operator, and we finally have the tools to do so. We can define the join in terms of indMax on the underlying trees. The proof that the indMax is idempotent on the result follows from associativity, commutativity, and monotonicity of indMax.

```
unfolding indMax Z ↑ indMaxView
max : SMBTree → SMBTree
\max t_1 t_2 =
  MkTree
   (indMax (rawTree t_1) (rawTree t_2))
   (indMax-swap4
     Brouwer. \leq \frac{9}{9} ind Max-mono (isldem t_1) (isldem t_2))
```

For Brouwer trees, indMax had all the properties we wanted except for idempotence. All of these can be lifted directly to SMB-trees:

 $\max - \leq L : \forall \{t_1 \ t_2\} \longrightarrow t_1 \leq \max t_1 \ t_2$ 

```
\max \le R : \forall \{t_1 \ t_2\} \longrightarrow t_2 \le \max t_1 \ t_2
                                                                                                                                        853
\mathsf{max\text{-}mono}: \forall \ \{t_1 \ t_1' \ t_2 \ t_2'\} \longrightarrow t_1 \le t_1' \longrightarrow t_2 \le t_2' \longrightarrow
                                                                                                                                        854
   \max t_1 \ t_2 \leq \max t_1' \ t_2'
                                                                                                                                        855
                                                                                                                                        856
\max\text{-idem} \le : \forall \{t\} \longrightarrow t \le \max t \ t
                                                                                                                                        857
max-commut : \forall t_1 t_2 \rightarrow max t_1 t_2 \leq max t_2 t_1
                                                                                                                                        858
\mathsf{max-assocL} : \forall \ t_1 \ t_2 \ t3 \longrightarrow \mathsf{max} \ t_1 \ (\mathsf{max} \ t_2 \ t3) \leq \mathsf{max} \ (\mathsf{max} \ t_1 \ t_2) \ t3 \overset{\mathsf{859}}{\sim}
\max-assocR: \forall t_1 t_2 t_3 \rightarrow \max(\max t_1 t_2) t_3 \leq \max t_1 (\max t_2 t_3)^{861}
```

In particular, max is strictly monotone, and distributes over the successor:

```
max-strictMono : \forall \{t_1 \ t'_1 \ t_2 \ t'_2 : SMBTree\}
   \rightarrow t_1 < t_1' \rightarrow t_2 < t_2' \rightarrow \max t_1 t_2 < \max t_1' t_2'
max-sucMono : \forall \{t_1 \ t_2 \ t_1' \ t_2' : SMBTree\}
   \rightarrow \max t_1 t_2 \leq \max t_1' t_2' \rightarrow \max t_1 t_2 < \max (\uparrow t_1') (\uparrow t_2')
```

However, because we restricted SMB-trees to only contain Brouwer trees that indMax is idempotent on, we can prove that Max is idempotent for SMB-trees:

```
max-idem : \forall \{t : SMBTree\} \rightarrow max \ t \ t \le t
max-idem \{t = MkTree \ t \ pf\} = mk \le pf
```

These together are enough to prove that our maximum is the least of all upper bounds.

```
\mathsf{max-LUB} : \forall \{t_1 \ t_2 \ t\} \longrightarrow t_1 \le t \longrightarrow t_2 \le t \longrightarrow \mathsf{max} \ t_1 \ t_2 \le t
max-LUB lt1 lt2 = max-mono lt1 lt2 ≤ \( \circ \) max-idem
```

Perhaps surprisingly, this means that an SMB-tree version of limMax is equivalent to max, since they are both the least upper bound:

```
\mathbb{N} \text{Lim} : (\mathbb{N} \to \mathsf{SMBTree}) \to \mathsf{SMBTree}
886
           NLim f = \text{Lim } CN \ (\lambda \ cn \rightarrow f \ (\text{Iso.fun } CN \text{Iso.fun}))
887
888
          max': SMBTree → SMBTree
889
          max' t_1 t_2 = NLim (\lambda n \rightarrow if0 n t_1 t_2)
890
891
          \max' - \leq L : \forall \{t_1 \ t_2\} \longrightarrow t_1 \leq \max' t_1 \ t_2
892
          \max' - \leq R : \forall \{t_1 \ t_2\} \longrightarrow t_2 \leq \max' t_1 \ t_2
893
894
          \max'-LUB: \forall \{t_1 \ t_2 \ t\} \rightarrow t_1 \le t \rightarrow t_2 \le t \rightarrow \max' t_1 \ t_2 \le t
895
896
          \max \le \max' : \forall \{t_1 \ t_2\} \longrightarrow \max t_1 \ t_2 \le \max' t_1 \ t_2
897
          max \le max' = max-LUB \ max' - \le L \ max' - \le R
898
          \max' \leq \max : \forall \{t_1 \ t_2\} \longrightarrow \max' \ t_1 \ t_2 \leq \max \ t_1 \ t_2
899
          max' \le max = max' - LUB max - \le L max - \le R
```

**4.2.4 Well Founded Ordering on SMB-trees.** Our motivation for defining SMB-trees was defining well founded recursion, so the final piece of our definition is a proof that the strict ordering of SMB-trees is well founded. Intuitively this should hold: if there are no infinite descending chains of Brouwer trees, and there are fewer SMB-trees than Brouwer trees, then there can be no infinite descending chains of SMB-trees. The key lemma is that an SMB-tree is accessible if its underlying Brouwer tree is.

```
sizeWF: WellFounded _{<}
sizeWF t = sizeAcc (Brouwer.ordWF (rawTree t))
where
sizeAcc: \forall \{t\}
\rightarrow Acc Brouwer._{<} (rawTree t)
\rightarrow Acc _{<} t
sizeAcc \{t\} (acc x)
= acc ((\lambda y lt \rightarrow sizeAcc (x (rawTree <math>y) (get \le lt))))
```

Thus, we have an ordinal type with limits, a strictly monotone join, and well founded recursion.

#### References

Guillaume Allais, Edwin Brady, Nathan Corbyn, Ohad Kammar, and Jeremy Yallop. 2023. Frex: dependently-typed algebraic simplification. arXiv:2306.15375 [cs.PL]

Yves Bertot and Pierre Castéran. 2004. Interactive Theorem Proving and Program Development. Springer-Verlag.

Marc Bezem and Thierry Coquand. 2022. Loop-checking and the uniform word problem for join-semilattices with an inflationary endomorphism. Theoretical Computer Science 913 (2022), 1–7. https://doi.org/10.1016/j.tcs.2022.01.017

Jonathan H.W. Chan. 2022. Sized dependent types via extensional type theory. Master's thesis. University of British Columbia. https://doi. org/10.14288/1.0416401 Nathan Corbyn. 2021. Proof Synthesis with Free Extensions in Intensional Type Theory. Technical Report. University of Cambridge. MEng Dissertation

Nicolai Kraus, Fredrik Nordvall Forsberg, and Chuangjie Xu. 2023. Type-theoretic approaches to ordinals. Theoretical Computer Science 957 (2023), 113843. https://doi.org/10.1016/j.tcs.2023.113843

The Univalent Foundations Program. 2013. Homotopy Type Theory: Univalent Foundations of Mathematics. https://homotopytypetheory.org/book, Institute for Advanced Study.